# QMAKE Configuration file
# Copyright (c) 2007-2010, WOSH - Wide Open Smart Home 
# by Alessandro Polo - OpenSmartHome.com
# All rights reserved.
#######################################################
# Revision $Id: persistence.pro 2502 2010-05-22 01:35:35Z alex $
#######################################################

DEFINES += WOSH_PERSISTENCE

!CONFIG(WOSH_PERSISTENCE_QT) {
	!CONFIG(WOSH_PERSISTENCE_libxml2) {
		CONFIG += WOSH_PERSISTENCE_QT
		message( "WOSH - FRAMEWORK [Persistence] (Unset, forced: Qt)" )
	}
}

#TEMP: force inclusion
CONFIG += WOSH_PERSISTENCE_plaintext
CONFIG += WOSH_PERSISTENCE_RAW

#####################################################

HEADERS += framework/persistence/EncoderProtocolXml.h
SOURCES += framework/persistence/EncoderProtocolXml.cpp

HEADERS += framework/persistence/EncoderProtocolSql.h
SOURCES += framework/persistence/EncoderProtocolSql.cpp

#####################################################

WOSH_PERSISTENCE_QT {
	# inherited by CONFIG : DEFINES += WOSH_PERSISTENCE_QT
	!include( framework/persistence/qt/persistence.qt.pro ) {
		error( "FATAL: Project file 'framework/persistence/qt/persistence.qt.pro' is missing! (disaligned distribution?)" )
	}
}

#####################################################

WOSH_PERSISTENCE_libxml2 {
	# inherited by CONFIG : DEFINES += WOSH_PERSISTENCE_libxml2
	!include( framework/persistence/libxml2/persistence.libxml2.pro ) {
		error( "FATAL: Project file 'framework/persistence/libxml2/persistence.libxml2.pro' is missing! (disaligned distribution?)" )
	}
}

#####################################################

WOSH_PERSISTENCE_plaintext {
	# inherited by CONFIG : DEFINES += WOSH_PERSISTENCE_plaintext
	!include( framework/persistence/plaintext/persistence.plaintext.pro ) {
		error( "FATAL: Project file 'framework/persistence/plaintext/persistence.plaintext.pro' is missing! (disaligned distribution?)" )
	}
}

#####################################################

WOSH_PERSISTENCE_RAW {
	DEFINES += WOSH_PERSISTENCE_RAW
	!include( framework/persistence/raw/persistence.raw.pro ) {
		error( "FATAL: Project file 'framework/persistence/raw/persistence.raw.pro' is missing! (disaligned distribution?)" )
	}
}

#####################################################

